Time-Dependent Covariates "Survival" More in PROC PHREG

Fengying Xue,Sanofi R&D, China
Michael Lai, Sanofi R&D, China

ABSTRACT

Survival analysis is a powerful tool with much strength, especially the semi-parametric analysis of COX model in PHREG, the most popular one. How to explain its enormous popularity? The most important reason is that it does not require you to choose some particular probability distribution under the proportional-hazards assumption. While in some cases, such as in long-term follow-up study or some covariates whose attribute really change over process (such as age, salary), the proportional-hazards assumption of constant hazard ratio is frequently violated, and PHREG can also make it. This is the second reason; it is relatively easy to incorporate time-dependent covariates. It provides the chance to modulate dynamic design, leading to a more robust and accurate outcome.

INTRODUCTION

We begin by defining a time-dependent variable and use Stanford heart transplant study as example. We also state the general formula for Cox model and how the Cox proportional hazards (PH) model can be extended to allow time-dependent variables, followed by a discussion bases on Stanford heart transplant study, including a description of the hazard ratio, two methods to handle time-dependent variable in PHREG. At last, we will check PH assumption by using multiple methods for accuracy and robustness.

BRIEF BACKGROUND

Time-dependent covariates are those that may change in value for a given subject over the course of observation. In contrast, time-independent covariates are those whose value remains constant over time. Take Stanford heart transplant study as example, regarding covariate TRANS (which is equal to 1 if the patient has already had a transplant at time t and 0 otherwise). In this example, we see that TRANS is time-dependent variable whose value changed after transplanted at time t_0 for the transplant one. While for the non- transplant one, its value is always 0. All in all, TRANS is a time-dependent variable for this study.



Based on attribute and reason of the value change, time-dependent variables can be classified into two categories: Internal and Ancillary. For internal variable, whose value change because of "internal" characteristics or behavior of the individual while ancillary variable whose values change because of "external" characteristics. An example of an ancillary variable is air pollution index at time t for a particular geographical area. Additionally, for above Stanford heart transplant, TRANS, it is part internal and part ancillary variable, individual traits, eligible transplant criteria, is internal and available donor is ancillary.



Why time-dependent analysis is so powerful in survival analysis? Firstly, since that outcome or endpoint (time to event) is a time related variable, if the explanatory variable is time-dependent variable it is probability that it is heavily correlated with each other. In such instances, it is easy to confuse the "causation" of effect and misleading the result. Secondly, some variables that are important to the risk of an endpoint will vary in individuals over the course of a study and there is no way to control them (i.e. keep them constant) by controlling the people's risk factors, such as people's alcohol related drinking habits, meat intake, and dairy intake. Additionally, FDA has asked for clarifying for "what estimated", also citing "*De facto*" estimate as their interest, so we must be aware of this kind of variables, time-dependent. If not, treat dependent as independent, it may cause bias in the estimation, even more incorrect inference regardless of significance of effects, and it may over fit model and cost much extraneous time and without estimate improvement. So let's extent PH COX model to extended COX model, time-depend COX model.

While it's simple to modify COX model for time-depend covariates, to modify PH COX model to included time-dependent covariates, we need to do is write (t) after the X_j (j>=p2) which is time dependent vector.

	PH COX model	Time-dependent model
define	Value of variable is constant over time	Value of variable differs over time
function	$h\ (t, \boldsymbol{X}(\boldsymbol{t})) = h_0 exp(\sum_{i=1}^{p_1} \beta_i X_i)$	$h\ (t, \boldsymbol{X}(\boldsymbol{t})) = h_0 exp(\sum_{i=1}^{p_1} \beta_i X_i + \sum_{j=1}^{p_2} \delta_j X_j\ (t))$

Where

$$\mathbf{X}(t) = \underbrace{(X_1, X_2, \ldots X_{p_1},}_{\text{Time-independent}} \underbrace{X_1(t), X_2(t), \ldots X_{p_2}(t))}_{\text{Time-dependent}}$$

Hazard function changed, so easy, meanwhile its hazard ratio also chaged as below.
X_k time-independent variable, HR is determinded by differnece of covariates value and its coefficient.

$$HR = \frac{h(t|X_{k1})}{h(t|X_{k2})} = \frac{h_0(t) exp(\sum_1^{k-1} \beta_i X_i + \beta_k X_{k1} + \sum_{k+1}^{p_1} \beta_i X_i)}{h_0(t) exp(\sum_1^{k-1} \beta_i X_i + \beta_k X_{k2} + \sum_{k+1}^{p_1} \beta_i X_i)} = exp(\beta_k (X_{k1} - X_{k2}))$$

X_k time-dependent variable, HR is depenterminded by differnece of covariates value at t and coefficient.

$$HR = \frac{h(t|X_{k1})}{h(t|X_{k2})} = \frac{h_0(t) exp(\sum_1^{p1} \beta_i X_i + \sum_1^{k-1} \beta_j X_j(t) + \beta_k X_{k1}(t) + \sum_{k+1}^{p_2} \beta_j X_j(t))}{h_0(t) exp(\sum_1^{p1} \beta_i X_i + \sum_1^{k-1} \beta_j X_j(t) + \beta_k X_{k2}(t) + \sum_{k+1}^{p_2} \beta_j X_j(t))} = exp(\beta_k (X_{k1}(t) - X_{k2}(t)))$$

Following we use one example to demostrate how to estimte time-dependent covarites' meaning, how to assess PH assumption. In this paper, we suppose that models fit well and no need to consider its reliability, for its model fitness, it is beyond of this scope.

TIME DEPENDENT MODEL

STANFORD HEART TRANSPLANT STUDY

We would like to use Stanford heart transplant study as example, not only because it's popular to everyone but also because it's clarify to explain its meaning and then we can draw inferences about other cases from this case. It is part internal and part ancillary time-dependent variable. Regardless of variable type, the form of extended Cox model and procedures for analysis are the same. Knowing its type well, it will be more accurate to know model well and provide scientific explanation.

As reported by Crowley and Hu (1977), the famous Stanford Heart Transloant Data, sample consistents of 103 cardiac patients who enrolled in the trasplanation program between 1976 and 1974. After enrollment patients waited vary lengths for time until a suitable donor heart was found. 30 patients died before receiving transplantation, while another 4 patietns had still not received transplants at the termination date of April 1, 1974. Patients were followed until death or the termination date. Of the 69 transplant recepients, only 24 were still alive at terminantion. The goal of the study was to assess whether patients receiving transplantation survived longer than patients not receiving transplantation.

Following are raw variables,

DOB	Date fo birth
DOA	Date of acceptance into the program
DOT	Date of transplant
DLS	Date of last seen(Dead or censored)
DEAD	Coded 1 if dead at DLS; otherwise, coded 0
SURG	Coded 1 if patients had open-heard sugery prior to DOA; Others, coded 0

The following are derived varibles

SURV1	DLS-DOA
AGEACCPT	(DOA-DOB)/365.25
AGETRANS	(DOA-DOT)/365.25
WAIT	DOT-DOA
TRANS	Coded 1 if patients were with DOT ; Others, coded 0

MISLEADING OUTCOME

There is one misleading outcome which not takes time-dependent into consideration, treat TRANS as time-indepent variable. The traditional method is using PH COX model,$hazard\ function = h_0(t)exp(\beta1 * TRANS + \beta2 * SURG + \beta3 * AGEACCPT)$ to estimate the COX regression of SURV1 on trasplant status(TRANS), controlling for SURG and AGEACCPT, TRANS and SURG are all treated as time-independent variables.

```
proc phreg data=stan;
    class trans(ref=first) surg;
    model surv1*dead(0)=trans surg ageaccpt;
run;
```

The result as below output show very strong effects on both transplant status and age at acceptance. We saw that each additionally year of age at the time of acceptance is associated with a 6 percent increase in the hazard of death. On the other hand, the hazard for those who received transplantation is only about 18 percent of the hazard for those who do not. Or equivalenetly, those who did not receive transplans are about 4.5(1/0.181-1≈4.5) times more likely to die at any given timepoint.

Analysis of Maximum Likelihood Estimates

Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
trans	1	1	-1.70813	0.27860	37.5902	<.0001	0.181	trans 1
surg	0	1	0.42130	0.37098	1.2896	0.2561	1.524	surg 0
ageaccpt		1	0.05860	0.01505	15.1611	<.0001	1.060	

Perhaps the age effect maybe real, but the transplantation effect is almost surely an artifact. The main reason is that TRANS is actually consequence of the dependent variable: an early death prevent a patient from getting transplantation, or taking the reciprocal, the more time patient live, the more chance to receive transplantation, so it was difficult to determine if transplantation actually reduced the risk of death or if people who lived longer were more likely to receive a transplantation. Below plot shows the primary reason for this misleading model.

For transplant patients, transplant win time should contribute to TRANS effect not the total survival time. In this analysis, the total survival will be favor for transplant, and then enlarge transplant effect. In fact, wait time belong to non-transplant effect the same as for non-transplant, so it will be better to split total duration of transplant patients into two ones, one is for wait time period without transplantation, another is transplant win time with transplantation. While for non-transplant patients, they are only with wait-time period. Under this split, no matter Transplant or Non-transplant, they will be treated fairly. It will make the result more accurate. This is also the primary idea for time dependent method, split into many intervals, during which hazard is constant, and then give analysis on each intervals, and combine them give a final outcome, generally.



TIME-DEPENDENT METHOD

To handle time-dependent covariates, PROC PHREG has two methods but with very different ways of setting up the datasets and specifying the models. In programming statements method, there is one record per indiviual and the time-dependent covariates are then defined in programming statement that is part of the PROC PHREG step. In Count Process method, on the other hand, there may be multiple records for each individual, each record corresponding to an interval during which all the covariates remain constant. Once this special data set has been constructed, time-dependent covariates are treated as just like time-independent covariates in each interval. Mainwhile there is a different syntax needed to specify the time-dependent variable.

COUNTING PROCESS

Regarding counting prcess, it has two steps, the first step is construct a dataset with multiply records per patient, one record for each period during which all the covariates remain constant. The second step is use special syntax to estimate model.

For the first step, it requests more programming absolutely but freely you can do it if you can think of it.

```
data countp;
    set explod.stan;

    *it is for wait-time=0 pts,since that 2 pts transplant once he acceptance into the study;
    if wait=0 then wait=0.1;

    * it is for survival=0 pts,since that 1 pts died on their day of accpetance into the study;
    if surv1=0 then surv1=0.1;

    * data construction;
    if trans=1 then do;
         start=0;   stop=wait; trans=0; dead2=0;    output;
         start=wait;stop=surv1;trans=1; dead2=dead; output;
    end;
    else do;
         start=0; stop=surv1; trans=trans; dead2=dead; output;
    end;
run;
```

For this case, as highlight in the first red rectangle, it is for patients who received transplantion, it needs two records: one for the interval between acceptance and transplantation, wait-time, with TRANS=0 and DEAD2=0 and the second one is the interval between transplantation and either dead or censoring, transplant win time, with TRANS =1 and DEAD2=dead. Regarding the second red rectagle, it is for patients who did not receive transplanation. One record is enough since its transplantaion status remain the same without change. Take ID=4, 5 as example.

ID	DOA	DOT	DLS	DEAD	DUR	SURG	TRANS	WAIT	SURV1
4	03/28/1968	05/02/1968	05/05/1968	1	39	0	1	35	38
5	05/10/1968		05/27/1968	1	18	0	0	10000	17

ID	DOA	DOT	DLS	DEAD	DUR	SURG	TRANS	WAIT	SURV1	START	STOP	TRANS(new)	DEAD2
4	03/28/1968	05/02/1968	05/05/1968	1	39	0	1	35	38	0	35	0	0
4	03/28/1968	05/02/1968	05/05/1968	1	39	0	1	35	38	35	38	1	1
5	05/10/1968		05/27/1968	1	18	0	0	10000	17	0	17	0	1

The second step is estimation by PHREG as below syntax. The only difference is that this syntax requires us to specify a starting and stopping time for each record. Interval (start, stop) stands for intervals in which covariaves are constant and during which the individual is continuously at risk and event only occurred at the end of intervals. It also reflects time-dependent variable if multiple intervals per person and covariate value change.

```
proc phreg data=countp;
     class trans(ref=first) surg;
     model (start, stop)*dead2(0)=trans surg ageaccpt;
run;
```

Under this model, it indicates that transplantation has no effect on the hazard of death. The effect of age at acceptance is somewhat smaller than before, it stationary relatively. It is different deeply from the previous traditionally one. This one is more accurate and more close the truth.

Analysis of Maximum Likelihood Estimates								
Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
trans	1	1	0.01442	0.30852	0.0022	0.9627	1.015	trans 1
surg	0	1	0.77160	0.35967	4.6022	0.0319	2.163	surg 0
ageaccpt		1	0.03053	0.01390	4.8262	0.0280	1.031	

PROGRAMMING STATEMENT

Programming statement is used to create or modify the values of the exlanaroty variables in the model statement. It is especially usefull in fitting models with time-dependent coavariabtes. Its syntax likes in data-step. Now let's see how we can use SAS to create a time-dependent covariate that changes for each individual from 0 to 1 at the time of the transplantation.

```
proc phreg data=explod.stan;
    class surg;
    model surv1*dead(0)=trans surg ageaccpt;
    if wait>=surv1 or wait=. then trans=0;else trans=1;
run;
```

Analysis of Maximum Likelihood Estimates								
Parameter		DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq	Hazard Ratio	Label
trans		1	0.01442	0.30852	0.0022	0.9627	1.015	
surg	0	1	0.77160	0.35967	4.6022	0.0319	2.163	surg 0
ageaccpt		1	0.03053	0.01390	4.8262	0.0280	1.031	

Perhaps there are two mysteries. The first is that a new variable is created, called 'TRANS' which equals 0 if the Waiting time is greater or equal than the survival time or if the waiting time is missing, otherwise TRANS=1. Does this make intuitive sense? The second is the result is magic, they are exactly the same with the result of count process. For the first question, the answer should be 'no', not intuitive, but it is correct. It will take a little effort to understand what this conditional statement in PHREG is doing.

The point is that the conditional statement in PHREG is handled differently than in a data step. If they are the same, programming statement can't be time-dependent data carrier. It's more complicated than simply evaluating whether WAIT>=SURV1 for each patient.

```
proc phreg data=stan;
    class surg;
    model surv1*dead(0)=trans surg ageaccpt;
    output out=out atrisk=atrisk;
    if wait>=surv1 or wait=. then trans=0;else trans=1;
run;
```

SURV1 is a variable that is evaluated at each event time point, a comparison is made between the event time and each patient's wait time. Based on those comparisons, each patient's duration was spited into multiply intervals, and the TRANS value will be 0 or 1, time-depend variable.

Let's use a plot to see its process clearly, red stars stand for survial starting or ending points, black stars stand for event time points and red heart stands for transplantation.



Its corrsponding tables as below, suppose Pt1 is transplant patient while Pt2 is not.
Transplant:

interval	start	end	condition	trans
Int 1	0	Event 1	Wait>Event 1	0
Int 2	Event 1	Event 2	Wait>Event 2	0
….	….	….	……	….
Int i	Event i-1	DOT	Wait=DOT	0
Int i+1	DOT	Event i	Wait<Event i+1	1
….	….	….	……	….
Int j	Event j-1	SURV1	Wait<Event j	1

Not Transplant:

interval	start	end	condition	trans
Int 1	0	Event 1	Wait>Event 1	0
Int 2	Event 1	Event 2	Wait>Event 2	0
Int 3	Event 3	Event 3	Wait>Event 3	0
….	….	….	……	….
Int l	Event l-1	SURV1	Wait> Event l	0

From above plots and tables we find that programming statement is the same as count process in fact, the difference is that programming statement split more interverls, if we combine all intervals by TRANS staus, we will got the same intervals as count process, so that's not surprise we got the same result from the two method.

There is another confusion, why assign 0.1 for patients who are with WAIT=0 or SURV1=0? Below are three special patients.

ID	DOA	DOT	DLS	DEAD	DUR	SURG	TRANS	WARIT	SURV1
3	01/06/1968	01/06/1968	01/21/1968	1	16	0	1	0	15
45	01/05/1971	01/05/1971	02/18/1971	1	45	0	1	0	44
15	09/27/1968		09/27/1968	1	1	1	0	10000	0

Patient, ID=15, who was dead once he was accepted into this study, so survial time is 0. If we treated its interval as (0, 0], this patients will be excluded from analysis in counting process. But for him, even for this study, 0 is an event time, it will lead to L_1 was missed in below partial likelihood funciton. By the way, the minxium event time except this patient, is 1, so that we can assign a value, called μ , $\mu \in (0, 1)$, such as 0.1, it will give the chance for patient ID=15 to be an element of partial likelihood function and this is the fact.

$PL = L_1 \times L_2 \times \cdots \times L_j$, Where j is the event sequence by survival time.

While for another two patients who took transplantation once they accepted into the study, and they were at risk or not during (0, μ], there are two and only two possible.

(1) if we suppose their wait time ≥ μ, they were not transplanted during (0, μ], TRANS=0
(2) if we suppose their wait time <μ, they were transplanted during (0, μ], TRANS=1

In mathematical logic, any of them is possible, we are not sure. The two will lead to different results. For (1), it is the same as the previeous example, as assign 0.1 for all three cases. It is caused by SURV1 & WAIT definition (before), without plus 1, so 0 appeared as survival time or wait time. If we use Define (after), plus one day for SURV1 and WAIT, it will avoid this issue and get the same results. It is because that its assessment is the same regardless based on value or ranked value due to the partial likelihood function, baseline hazard function is canceled out, depend only on the ranks of the event times not their numerical values. This implies that any monotonic transformation of the event times will leave the coefficient estimates unchanged. So, WAIT=0 and SURV1=0, with same ranked value, assign= μ ,such as 0.1. This is the whole history.

Variable	Define(before)	Define(after)
SURV1	DLS-DOA	DLS-DOA+1
WAIT	DOT-DOA	DOT-DOA+1

As shown, the counting process requires substantially more coding. However, this upfront effort makes it easier to detect and correct errors because a data set is created and can be debugged. The programming statements are faster to code and can handle special case, but the coding seems to be tricky and there is no way to detect if the time-varying covariates have been coded correctly. Furthermore, when using the programming statement method, a temporary dataset containing the time-varying covariates has to be created each time PROC PHREG is run. Depending on how large the dataset is, this could drastically increase computing time. For these reasons, we prefer the counting process. We also more prefer to use one to validate another one for robustness.

Our results from the examples illustrated how impactful time-dependent variables can be in Cox PH model. When we used static variables only in the model, TRANS had effect on SURV1. But when we used more time points of TRANS status, we saw a very insignificant effect of TRANS on SURV1. So it is strengthful to use time-dependent variables if it is, meanwhile it should be caution to use time-dependent modle since that if not it will cause much time in running model and cost much time to investigate and demonstrate it. So before we use time-dependent model, we must test its PH assumption and search for clinical expert suggestions.

ASSESS PH ASSUMPTION

There are three general approaches for assessing the PH assumption, again listed here, graphical, goodness of fit and time-dependent variables

GRAPHICAL

We now briefly overview each approach, starting with graphical techniques. Totally, 2 types of plots are involved. The quick and easy one is the Kaplan-Meier plot which estimates of the survival function to checks of proportional hazards. If proportional hazards hold, the graphs of the surival function should look "parallel", in the sense that they should have basically the same shape, should not cross, and should start close and then diverge slowly through follow up time.



```
proc lifetest data=explod.stan METHOD=km plots=survival( atrisk(OUTSIDE(0.1) maxlen=15));
    strata trans;
    time surv1*dead(0);
    format trans trans.;
    label trans='Legend:';
run;
```

There is another type of graphical technique available. The mose popular one, it involves comparing estimated ln(–ln) survivor curves over different (combinations of) categories of variables being investigated. Based on below calculation, Parallel curves indicate that the PH assumption is satisfied.

$$S(t) = \exp\big(-H(t)\big)$$

$$-logS(t) = -log(exp\big(-H(t)\big)) = -(-H(t)) = H(t)$$

$$log\big(-logS(t)\big) = logH(t) = log\left(\int_0^t h(u)du\right)$$

$$= log\left(\int_0^t \boldsymbol{\lambda_0}(u)\, exp\big(\boldsymbol{\beta_1 X_1} + \boldsymbol{\beta_2 X_2}(u)\big)\, du\right)$$

$$\xrightarrow{PH\ retrived\ ,\ \boldsymbol{\beta_2 X_2}(t)=0} log\left(\int_0^t \boldsymbol{\lambda_0}(u)\, exp(\boldsymbol{\beta_1 X_1})\, du\right) = log\,\Lambda(t) + \boldsymbol{\beta_1 X_1}$$

$$\xrightarrow{\textbf{\textit{distance between two surival function}}} (log\,\Lambda(t) + \beta_1 X_{1i}) - \big(log\,\Lambda(t) + \beta_1 X_{1j}\big) = \beta_1 (X_{1i} - X_{1j})$$

```
proc lifetest data=stan method=km plots=(lls );
    strata trans;
    time surv1*dead(0);
    format trans trans.;
run;
```



From plot methods, it is easy to get an idea, but the big issue is that how 'Parallel' is parallel? From above two kinds of plots, it seems that we can find any time-dependent clue. But how 'Parallel' is Parallel? We would like to find one value with statistical meaning. Beyond this method, we have alternative method to test PH assumption.

GOODNESS OF FIT

A famous method for evaluating PH assumption is to examine the Schoenfeld residuals. The Schoenfeld residual is defined as the difference between covariate for observation and the weighted average of the covariate values for all subjects still at risk when observation experiences the event.

$$r_{Sik} = x_{ik} - \sum_{j=1}^{n_i} x_{jk} p_j$$

For covariate x_k, it is the difference between the i_{th} individual's covariate, x_{ik}, and the "expected" value of x_k for all people at risk, where p_j is the probability that person j had event at time t_i. Schoenfeld residuals are mainly used to detect departures from the proportional hazards assumption. If there is a pattern in these residuals against survival time, the PH assumption is questionable. If PH satisfied, plots will show no trend over time and its average is 0 across time.

Additionly, we can also take expected function, It can be shown that:

$$E[r_{Sik}] \approx \beta_k(t_i) - \hat{\beta}_k$$

Where $\beta_k(t_i)$ is a time dependent coefficient (or, alternatively, the corresponding covariate is time dependent). We also assess PH from β coefficent.

Firstly, we would like to use residual plot to get a rough idea. This can be attrived by Proc PHREG to get Schoenfeld residual.

```
proc phreg data=stan;
    class surg;
    model surv1*dead(0)=trans surg ageaccpt;
    output out=resid ressch=schtrans schsurg schage;
run;
```

And then get its residual trend with according survival time,

```
proc loess data = resid;
    model schtrans=surv1 / smooth=(0.2 0.4 0.6 0.8);
run;
```



From above plot, we find that lack of residuals for long survival times for the 'trans' covariate as in red ellipses, it indicates a possible time dependent coefficient. As it turns out, a time dependent covariate analysis was required for this dataset. This is just a plot impression, we would like more to see its relationship, since that the idea behind the statistical test is that if the PH assumption holds for a particular covariate then the Schoenfeld residuals for that covariate will not be related to survival time. So it is better to get its relation coefficient and its P value to test significance or not.

```
proc rank data=resid out=resrank ties=mean;
    var surv1;
    ranks surv1_rank;
run;
```

```
proc corr data=resrank ;
    var schtrans;
    with surv1_rank;
run;
```

And the result is as below.

Pearson Correlation Coefficients Prob > |r| under H0: Rho=0 Number of Observations	
	schtrans
surv1_rank	0.29640
	0.0098
Rank for Variable surv1	75

```
proc reg data=resrank;
    model surv1_rank=schtrans;
run;
```

And the result is as below,

Parameter Estimates						
Variable	Label	DF	Parameter Estimate	Standard Error	t Value	Pr > |t|
Intercept	Intercept	1	42.54662	2.94244	14.46	<.0001
schtrans	Schoenfeld Residual for trans	1	19.07863	7.19523	2.65	0.0098

From Schoenfeld residual plot and its rank test, we see that Schoenfeld residual correlation with survival time, PH assumption violated significantly at P=0.01 level.

There is another good of fitness method for PH assessment via Proc PHREG also. The procedure, developed by Lin, Wei, and Zing (1990), can detect violations of proportional hazards by using a transform of the Schoenfeld residuals known as the empirical score process. The empirical score process under the null hypothesis of no model misspecification can be approximated by zero mean Guassian processes, and the observed score process can be compared to the simulated processes to asses departure from proportional hazards.

It is easy to execute, adding **ph** options on the assess statement. All the exploratory variables will be assessed this way. We also specify the resample option, which performs a supremum test of the null hypothesis that PH assumption holds. Essentially, the supremum tests calculate the proportion of 1000 simulations that contain a maximum standardized score larger than the observed maximum standardized score. This proportion is reported as the p-value. If only a small proportion, say 0.05, of the simulations have a standardized score larger than the observed maximum, then that suggests that violation of proportional hazard.

```
proc phreg data=stan ;
    class surg;
    model surv1*dead(0)=trans surg ageaccpt;
    assess ph /resample ;
run;
```

Let's look at the model with just a linear effect for TRANS.

Supremum Test for Proportionals Hazards Assumption				
Variable	Maximum Absolute Value	Replications	Seed	Pr > MaxAbsVal
trans	1.5854	1000	968135960	0.0020



We find that the solid lines represent the observed standardized score, while dotted lines represent 20 simulated sets of standardized score under the null hypothesis that PH assumption holds. A solid line that falls significantly outside the boundaries set up collectively by the dotted lines suggest that observed standardized score do not conform to the expected standardized score. This graph look particularly alarming and the supremum tests are significant, suggesting that PH assumption violated.

TIME-DEPENDENT VARIABLES

There is still another method to test PH assumption, it is just like time-dependent method, putting time related (suppose) variable into PH COX model to test its significient use -2 L test. This is simple but with more clinical science and so without additionally specify in this paper.

CLONCLUSION

In this paper we introduce extended COX model via PH COX model and then have shown an example with two methods used to handle time-varying covariates, count process and programming statement. We prefer the counting process. We prefer to use one to validate another one for robustness. It is strengthful to use time-dependent variables if it is, meanwhile it should be caution to use time-dependent model since that if not it will cause much time in running model and cost much time to investigate and demonstrate it. So before we use time-dependent model, we must test it PH assumption and search for clinical expert suggestion. We also introduce some methods to check PH assumption, it is important, and will favor an object outcome. So before using time-dependent model, we must test PH assumption and ask for expert's suggestion in case of over fitting model or under estimate model. Additionly, they can not do without PROC PHREG, a very useful tool.

REFERENCE

M. Gail, K. Krickeberg, J.M. Samet, A. Tsiatis, W. Wong(2012) Survival Analysis A Self-Learning Text Third Edition
Allison, P. D. (2010) Survival Analysis Using SAS: A Practical Guide: Sas Inst
Teresa M. Powell, Melissa E. Bagnell(2012), Your “Survival” Guide to Using Time‐Dependent Covariates
Mark Jones (2016), Time-dependent bias in observational studies of oseltamivir
SAS Seminar (2014) Introduction to Survival Analysis in SAS

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Fengying Xue
Enterprise: Sanofi R&D China
Address: 2th Floor, HNA Plaza, No 108, Jianguo Road, Chaoyang District
City, State ZIP: Beijing, 100022
Work Phone: 86 10 6563 4915
E-mail: Fengying.xue@sanofi.com

Name: Michael Lai
Enterprise: Sanofi R&D China
Address: 2th Floor, HNA Plaza, No 108, Jianguo Road, Chaoyang District
City, State ZIP: Beijing, 100022
Work Phone: 86 10 6563 4900
E-mail: Michael.Lai@sanofi.com